
Abstract
An LL(1) parser is a recursive descent algorithm that uses a single token of lookahead to build a
grammatical derivation for an input sequence. We present an LL(1) parser generator that, when
applied to grammar G, produces an LL(1) parser for G if such a parser exists. We use the Coq Proof
Assistant to verify that the generator and the parsers that it produces are sound and complete, and
that they terminate on all inputs without using fuel parameters. As a case study, we extract the
tool’s source code and use it to generate a JSON parser. The generated parser runs in linear time; it
is two to four times slower than an unverified parser for the same grammar.

1

Introduction

Many software systems employ parsing techniques to map sequential input to structured
output. Often, a parser is the system component that consumes data from an untrusted
source—for example, many applications parse input in a standard format such as XML or
JSON as the first step in a data-processing pipeline. Because parsers mediate between the
outside world and application internals, they are good targets for formal verification; parsers
that come with strong correctness guarantees are likely to increase the overall security of
applications that rely on them.
Several recent high-profile software vulnerabilities demonstrate the consequences of using
unsafe parsing tools. Attackers exploited a faulty parser in a web application framework,
obtaining the sensitive data of as many as 143 million consumers [5, 14]. An HTML parser
vulnerability led to private user data being leaked from several popular online services [6].
And a flaw in an XML parser enabled remote code execution on a network security device—a
flaw that received a Common Vulnerability Score System (CVSS) score of 10/10 due to its
severity [13]. These and other examples highlight the need for secure parsing technologies.

Parsing is a widely studied topic, and it encompasses a range of techniques with different
advantages and drawbacks [7]. One family of parsing algorithms, the top-down or LL-style
algorithms, shares several strengths relative to other strategies. LL parsers typically produce
clear error messages, and they can easily be extended with semantic actions that produce
user-defined data structures; in addition, generated LL parser code is often human-readable
and similar to hand-written code [15].
The common ancestor of the LL family is LL(1), a recursive descent algorithm that avoids
backtracking by looking ahead at a single input token when it reaches decision points. Its
descendants, including LL(k), LL(*), and ALL(*), share an algorithmic skeleton. Each of
these approaches comes with different tradeoffs with respect to expressiveness vs. efficiency.
For example, LL(1) operates on a restricted class of grammars and offers linear-time execution,
while ALL(*) accepts a larger class of grammars and runs in O(n4 ) time [16]. Different
algorithms are therefore suited to different applications; it is often advantageous to choose
the most efficient algorithm compatible with the language being parsed.
In this paper, we present Vermillion, a formally verified LL(1) parser generator. This tool
is part of a planned suite of verified LL-style parsing technologies that are suitable for a wide
range of data formats. We implemented and verified the parser generator using the Coq Proof
Assistant [19], a popular interactive theorem prover. The tool has two main components.
The first is a parse table generator that, when applied to a context-free grammar, produces
an LL(1) parse table—an encoding of the grammar’s lookahead properties—if such a table
exists for the grammar. The second component is an LL(1) algorithm implementation that
is parameterized by a parse table. By converting a grammar to a table and then partially
applying the parser to the table, the user obtains a parser that is specialized to the original
grammar. The paper’s main contributions are as follows:
1. End-to-End Correctness Proofs – We prove that both the parse table generator and
the parser are sound and complete. The generator produces a correct LL(1) parse table
for any grammar if such a table exists. The parser produces a semantic value for its input
that is correct with respect to the grammar used to generate the parser. Although prior
work has verified some of the steps involved in LL(1) parse table generation [2], to the
best of our knowledge, our LL(1) parse table generator and parser are the first formally
verified versions of these algorithms.
2. Total Algorithm Implementations – We prove that the parse table generator and
parser terminate on both valid and invalid inputs without the use of fuel-like parameters.
To the best of our knowledge, we are the first to prove this property about a parser
generator based on the context-free grammar formalism. Some existing verified parsers
are only guaranteed to terminate on valid inputs; others ensure termination by means
of a fuel parameter, which can produce “out of fuel” return values that do not clearly
indicate success or failure. A guarantee of termination on all inputs is useful for ruling
out denial-of-service attacks against the parser.
3. Efficient Extractable Code – We used Coq’s Extraction mechanism [10] to convert
Vermillion to OCaml source code and generated a parser for a JSON grammar. We then
used Menhir [17], a popular OCaml parser generator, to produce an unverified parser for
the same grammar and compared the two parsers’ performance on a JSON data set. The
verified parser was two to four times slower than the unverified and optimized one, which
is similar to the reported results for other certified parsers [8, 9]. Our implementation
empirically lives up to the LL(1) algorithm’s theoretical linear-time guarantees.
Along the way, we deal with several interesting verification challenges. The parse table
generator performs dataflow analyses with non-obvious termination metrics over context-free
grammars. To implement and verify these analyses, we make ample use of Coq’s tools for

defining recursive functions with well-founded measures, and we prove a large collection of
domain-neutral lemmas about finite sets and maps that may be useful in other developments.
The parser also uses well-founded recursion on a non-syntactic measure, and our initial
implementation must perform an expensive runtime computation to terminate provably;
in the final version, we make judicious use of dependent types to avoid this penalty while
still proving termination. Our parser completeness proof relies on a lemma stating that if
a correct LL(1) parse table exists for some grammar, then the grammar contains no left
recursion. Our proof of this lemma is quite intricate, and we were unable to find a rigorous
proof of this seemingly intuitive fact in the literature.
Our formalization consists of roughly 8,000 lines of Coq definitions and proofs. The
development is available at the URL listed as Supplement Material above.
This paper is organized as follows: in §2, we review background material on context-free
grammars and LL(1) parsing. In §3, we describe the high-level structure of our parse table
generator and its correctness proofs. In §4, we present the LL(1) parsing algorithm and its
correctness properties. In §5, we present the results of evaluating our tool’s performance on
a JSON benchmark. We discuss related work in §6 and our plans for future work in §7.

2

Grammars and Parse Tables

2.1

Grammars

Our grammars are composed of terminal symbols drawn from a set T and nonterminal
symbols drawn from a set N . Throughout this work, we use the letters {a, b, c} as terminal
names, {X, Y, Z } as nonterminal names, {s, s’, ...} as names for arbitrary symbols (terminals
or nonterminals), and {α, β, γ} as names for sentential forms (finite sequences of symbols).
A grammar consists of a start symbol S ∈ N and a finite sequence of productions P
(described in detail below). In addition, we require the grammar writer to provide a mapping
from each grammar symbol s to a type JsK in the host language (i.e., a Coq type). We
borrow this mapping from a certified LR(1) parser development [8]; it enables us to specify
the behavior of a parser that maps a valid input to a semantic value with a user-defined type,
rather than simply recognizing the input as valid or building a generic parse tree for it. The
symbols-to-types mapping supports the construction of flexible semantic values as follows:
The parser consumes a list of tokens, where each token is a dependent pair (a, v) of a
terminal symbol a and a semantic value v of type [a]. When the parser successfully
consumes a token (a, v), it produces the value v.
A production X → γ {f } consists of a left-hand nonterminal X, a right-hand sentential
form γ, and a semantic action f of type JγK → JXK. The notation JγK refers to the tuple
type built from the symbols in γ—for example, JaY K = JaK × JY K. After the parser uses
a production’s right-hand side to construct a tuple of type JγK, it applies f to this tuple
to produce a final semantic value of type JXK. The user provides semantic actions at
grammar definition time; these actions are dependently typed Coq functions. Throughout
this work, we use the notation X → γ to refer to a production when its semantic action
is clear from context or irrelevant to the discussion.

2.2

LL(1) Derivations

We define a derivation relation over a grammar symbol s, a word or token sequence w that s
derives, and a semantic value v that s produces for w. Because it is useful for a parser to
produce a semantic value for a prefix of its input sequence and return the remainder of the

sequence along with the value, the derivation relation also includes the remainder, or the
v
unparsed suffix of the input. The relation has the judgment form s −
→ w | r, which is read,
“s derives w, producing v and leaving r unparsed.”
The derivation relation appears in Figure 1. It is mutually inductive with an analogous
relation (also in Figure 1) over a list of symbols γ, a word w, a tuple of semantic values vs,
vs
and a remainder r. This second relation has the judgment form γ =
⇒ w | r (“γ derives w,
producing vs and leaving r unparsed”).
DerNT

X → γ {f } ∈ P
peek(w ++ r) ∈ LOOKAHEAD(X → γ)
vs
γ=
⇒w|r

DerT
v

a−
→ (a, v) | r

f vs

X −−−→ w | r
DerCons

vs

v

DerNil

s−
→ w | w0 ++ r

γ=
⇒ w0 | r

(v, vs)

()

[] =
⇒|r

s :: γ ====⇒ w ++ w0 | r

Figure 1 Derivation relations for symbols and lists of symbols.

The DerNT rule is the only LL(1)-specific rule in the relation. The peek function returns
a value l ∈ T ∪ {EOF} that is either the first token of the input sequence w ++ r, or EOF
if the entire sequence is empty. The rule itself states that production X → γ {f } applies
when peek(w ++ r) and X → γ are in the LOOKAHEAD relation (Figure 5)—i.e., when the
first input token “predicts” that production. To make this lookahead concept precise, we
introduce the definitions of several predicates that are commonly used in parsing theory to
relate a grammar’s structure to its semantics.

2.3

NULLABLE, FIRST, and FOLLOW

A nullable grammar symbol is a symbol that can derive the empty word . The NULLABLE
relation (Figure 2) captures the syntactic pattern that makes a symbol nullable. A nonterminal
is nullable if it appears on the left-hand side of a production and every symbol on the righthand side is also nullable (note that an empty right-hand side makes the left-hand nonterminal
trivially nullable). A sentential form γ is nullable if it consists entirely of nullable symbols.
We overload our notation for nullable symbols, writing NULLABLE(γ) to represent the fact
that γ is a nullable symbol sequence.
NuSym

X → γ {f } ∈ P

NULLABLE(γ)

NULLABLE(X)

NuGamma

∀i ∈ {1...n}, NULLABLE(si )
NULLABLE(s1 ...sn )

Figure 2 NULLABLE relation.

The FIRST relation (Figure 3) for a symbol s describes the set of terminals that can begin
a word derived from s. If s derives a word beginning with terminal a, then a ∈ FIRST(s).
Once again, we extend this concept to sentential forms, writing a ∈ FIRST(γ) if γ derives a
word that begins with a.

FirstNT

FirstT

a ∈ FIRST(γ)

X → γ {f } ∈ P

a ∈ FIRST(a)

a ∈ FIRST(X)

FirstGamma

NULLABLE(α)

a ∈ FIRST(s)

a ∈ FIRST(αsβ)
Figure 3 FIRST relation.

The FOLLOW relation (Figure 4) for a symbol s describes the set of terminals that can
appear immediately after a word derived from s. There is a standard practice among parser
implementers of placing the EOF symbol in FOLLOW(S), where S is the start symbol, so
that the parser can consume the entire input sequence. We follow this practice by adding
the FollowStart rule to the relation.
FollowStart

S is the start symbol

EOF ∈ FOLLOW(S)
FollowLeft

X → αY β {f } ∈ P

FollowRight

X → αY β {f } ∈ P

a ∈ FIRST(β)

a ∈ FOLLOW(Y )
NULLABLE(β)

l ∈ FOLLOW(X)

l ∈ FOLLOW(Y )
Figure 4 FOLLOW relation.

With these definitions in hand, we can give a precise definition for the judgment form
l ∈ LOOKAHEAD(X → γ) (“l is a lookahead token for production X → γ”) in Figure 5.
Intuitively, l is a token that, when it begins a sequence ts, “predicts” that the production
can derive a prefix of ts. As a special case, if the production derives ts = , then EOF ∈
LOOKAHEAD(X → γ). When an LL(1) parser builds a derivation from nonterminal
X for a prefix of ts, it “looks ahead” at ts and applies a production X → γ such that
peek(ts) ∈ LOOKAHEAD(X → γ).
FirstLk

l ∈ FIRST(γ)

l ∈ LOOKAHEAD(X → γ)

FollowLk

NULLABLE(γ)

l ∈ FOLLOW(X)

l ∈ LOOKAHEAD(X → γ)

Figure 5 LOOKAHEAD relation.

2.4

Parse Tables

An LL(1) parse table is a data structure that encodes a grammar’s lookahead information. An
LL(1) parser uses a parse table as an oracle; it consults the table to choose which productions
to apply as it builds a derivation for a token sequence.

A parse table’s rows are labeled with nonterminals and its columns are labeled with
lookahead symbols. Its cells contain production right-hand sides. A cell at row X and column
l that contains γ, written (X, l) 7→ γ, represents the fact l ∈ LOOKAHEAD(X → γ).
Figure 6 contains a grammar and its LL(1) parse table. Cell (X, b), for instance, contains
Zc (the right-hand side of production 2) because of the fact b ∈ FIRST(Zc). Cell (Z, c)
contains Y (the right-hand side of production 5) because of the facts NULLABLE(Y ) and
c ∈ FOLLOW(Z).
(X is the start symbol)
1. X → aY
2. X → Zc

3. Y → 

4. Z → b
5. Z → Y

X
Y
Z

a
aY

b
Zc
b

c
Zc

Y

EOF


Figure 6 Example grammar and its LL(1) parse table.

A correct LL(1) parse table for grammar G contains all and only the lookahead facts
about G—i.e., (X, l) 7→ γ ⇐⇒ l ∈ LOOKAHEAD(X → γ). Not every grammar has a
correct LL(1) parse table. If l ∈ LOOKAHEAD(X → γ) and l ∈ LOOKAHEAD(X → γ 0 ),
where γ =
6 γ 0 , then no correct table exists for G—a parser would be unable to choose whether
to apply γ or γ 0 upon encountering nonterminal X and token l. A grammar that has a correct
LL(1) parse table is called an LL(1) grammar.

3

Parse Table Generator Correctness Properties and Verification

We now describe the process of developing and verifying an LL(1) parse table generator. Our
first goal is to define the Coq function parseTableOf : grammar -> sum error_message
parse_table. (A value of type sum A B is either inl A or inr B.) We then wish to prove
that the function is both both sound (every table that it produces is the correct LL(1) parse
table for its input grammar) and complete (it produces the correct LL(1) parse table for the
grammar if such a table exists).

3.1

Structure of Parse Table Generator

Many standard compiler references describe variations on an algorithm for constructing
an LL(1) parse table from a grammar. The algorithm typically involves computing the
grammar’s NULLABLE, FIRST, and FOLLOW sets, and then constructing the table from
these sets (or returning an error value if a table cell contains multiple entries, in which case
no correct parse table exists for the grammar). Appel’s Modern Compiler Implementation in
ML [1], for example, contains pseudocode for performing the first of these two steps. The
algorithm presents several interesting challenges from a verification standpoint:
1. It uses an “iterate until convergence” strategy to perform a dataflow analysis over the
grammar. Such an algorithm is difficult to implement in a total language because it has
no obvious (i.e., syntactic) termination metric.
2. NULLABLE, FIRST, and FOLLOW are all computed simultaneously, so a proof of the
function’s correctness must simultaneously deal with the correctness of all three sets.
It is also possible to perform the NULLABLE, FIRST, and FOLLOW dataflow analyses
sequentially (in that order) because each analysis depends only on the previous ones. This
sequential approach is preferable from a proof engineering perspective, because we can clearly

state the correctness criteria for each step and verify the implementation independently of
the other steps. It is also preferable from a code reuse perspective, because some individual
steps may be useful in the context of other developments (for example, many species of
parser generators need to compute the set of nullable nonterminals). Therefore, we structure
our parse table generator as a pipeline of small functions that perform the following steps:
(1)
(2)
(3)
(4)
(5)

Compute the set of nullable nonterminals.
For each nonterminal X, compute FIRST(X) (using NULLABLE).
For each nonterminal X, compute FOLLOW(X) (using NULLABLE and FIRST).
Using NULLABLE, FIRST, and FOLLOW, compute the set of parse table entries.
Build a table from the set of entries, or return an error if the set contains a conflict.

Several steps involve similar reasoning and require the same proof techniques. In the next
section, we examine step (1) and its correctness proof in detail to illustrate these techniques.

3.2

Implementation of NULLABLE Dataflow Analysis

The first step in the parse table generation process is to compute the set of nullable
nonterminals. Our goal is to define the function mkNullableSet : grammar -> NtSet.t
(where NtSet.t is the type of finite sets of nonterminals) and then prove that when this
function is applied to grammar g, the resulting set contains all and only the nullable
nonterminals from g. We formalize this correctness property and theorem statement in Coq
as follows (nullable_sym is the mechanized version of the NULLABLE relation in Figure 2):
Definition nullable_set_correct (nu : NtSet.t) (g : grammar) :=
forall (x : nonterminal), NtSet.In x nu <-> nullable_sym g (NT x).
Theorem mkNullableSet_correct :
forall (g : grammar), nullable_set_correct (mkNullableSet g) g.
Portions of the mkNullableSet implementation appear in Figure 7. We represent a
grammar as a record with fields start : nonterminal and prods : list production.
The expression g.(prods) projects the prods field from a grammar. The auxiliary function
mkNullableSet’ takes a (possibly incomplete) NULLABLE set nu as an argument and
performs a single pass of the NULLABLE dataflow analysis over the grammar’s productions,
which produces a (possibly updated) set nu’. If nu has converged—i.e., if it is a fixed point
of the dataflow analysis—then it is returned. Otherwise, the algorithm performs another
iteration of the analysis, using nu’ as the starting point.
Because of this algorithm’s “iterate until convergence” structure, we need to do some
extra work to prove that it terminates. To accomplish this task, we use Coq’s Program
extension [18], which provides support for defining functions using well-founded recursion.
The Program Fixpoint command enables the user to define a non-structurally recursive
function by providing a measure—a mapping from one or more function arguments to a
value in some well-founded relation R—and then showing that the measure of recursive call
arguments is less than that of the original arguments in R.
In the case of mkNullableSet’, the measure (called countNullCands in Figure 7) is the
cardinality of nu’s complement with respect to the universe U of grammar nonterminals. We
then prove that if the NULLABLE set is different before and after a single iteration of the
analysis, then the more recent version contains a nonterminal that was not present in the
previous version, and therefore that the set’s complement with respect to U has decreased
(this fact is captured in the lemma nullablePass_neq_candidates_lt).

Lemma nullablePass_neq_candidates_lt :
forall (ps : list production) (nu : NtSet.t),
~ NtSet.Equal nu (nullablePass ps nu)
-> countNullCands ps (nullablePass ps nu) < countNullCands ps nu.
Program Fixpoint mkNullableSet’ (ps : list production) (nu : NtSet.t)
{ measure (countNullCands ps nu) } : NtSet.t :=
let nu’ := nullablePass ps nu in
if NtSet.eq_dec nu nu’ then nu else mkNullableSet’ ps nu’.
Next Obligation.
apply nullablePass_neq_candidates_lt; auto.
Defined.
Definition mkNullableSet (g : grammar) : NtSet.t :=
mkNullableSet’ g.(prods) NtSet.empty.
Figure 7 Selected portions of the mkNullableSet implementation.

Now that we have a suitable definition of mkNullableSet and a proof that it terminates,
we turn to the proofs of its main correctness properties.

3.3

Soundness of NULLABLE Analysis

One property of mkNullableSet that we wish to verify is that the function is sound—i.e.,
every nonterminal in the set that it returns really is nullable in g:
Definition nullable_set_sound (nu : nullable_set) (g : grammar) :=
forall (x : nonterminal), NtSet.In x nu -> nullable_sym g (NT x).
Theorem mkNullableSet_sound :
forall (g : grammar), nullable_set_sound (mkNullableSet g) g.
The soundness proof’s structure arises from the intuition that soundness holds not only of
mkNullableSet’s final return value, but of the intermediate sets that the function computes
along the way—in other words, soundness is an invariant of the function. We prove this
invariant with the following two lemmas:
(1) The initial set passed to mkNullableSet’ is sound
(2) If nu is sound, then mkNullableSet’ applied to nu is also sound
(1) is simple to prove, because the initial nu argument passed to mkNullableSet’ is the
empty set, which is trivially sound. Our earlier reasoning about the termination properties
of mkNullableSet’ pays dividends in the proof of (2), because we can proceed by wellfounded induction on the function’s measure. The main lemma involved in this proof states
that a single iteration of the dataflow analysis (called nullablePass in Figure 7) preserves
soundness of the NULLABLE set.

Completeness of NULLABLE Analysis

In addition to being sound, mkNullableSet should be complete—that is, every nullable
nonterminal from g should appear in the set that the function returns:
Definition nullable_set_complete (nu : NtSet.t) (g : grammar) :=
forall (x : nonterminal), nullable_sym g (NT x) -> NtSet.In x nu.
Theorem mkNullableSet_complete :
forall (g : grammar), nullable_set_complete (mkNullableSet g) g.
Once again, the proof is based on well-founded induction on the mkNullableSet’ measure.
In the interesting case, we must prove nu complete given the fact that nu and (nullablePass
g.(prods) nu) are equal. In other words, we need to show that any fixed point of the dataflow
analysis is complete. We isolate this fact in the lemma nullablePass_equal_complete:
Lemma nullablePass_equal_complete :
forall (g : grammar) (nu : NtSet.t),
NtSet.Equal nu (nullablePass g.(prods) nu)
-> nullable_set_complete nu g.
After some simplification, we are left with this goal:
nullable_sym g x

nu = nullablePass g.(prods) nu
NtSet.In x nu

The proof proceeds by induction on the nullable_sym judgment. Because this relation
is mutually inductive with nullable_gamma, we use Coq’s Scheme command to generate
a suitably powerful mutual induction principle for the two relations. Using this principle
requires some extra work because the programmer must manually specify the two properties
that the induction is intended to prove—one for symbols, and one for lists of symbols.
It can be difficult to come up with the right instantiations for mutual induction principles
such as this one. For several of the proofs in this development, such a choice was the most
difficult step. In some cases, we were able to avoid this problem by finding mutual inductionfree variants of relations whose pencil-and-paper definitions seem to call for mutuality.

3.5

Correctness of Parse Table Generator

Computing the NULLABLE set is the first of several dataflow analyses involved in generating
an LL(1) parse table. The correctness proofs for the remaining steps are similar in structure
to the NULLABLE proofs. For example, the FIRST and FOLLOW analyses each have
a soundness proof based on the fact that soundness is an invariant of the analysis, and a
completeness proof based on the fact that a fixed point of the analysis must be complete.
After proving each step correct given the correctness of previous steps, we can verify
parseTableOf—the function that implements the entire sequence—simply by chaining
together the proofs for the individual steps. The parseTableOf soundness and completeness
theorem statements appear below:
Theorem parseTableOf_sound :
forall (g : grammar) (tbl : parse_table),
parseTableOf g = inr tbl
-> parse_table_correct tbl g.

ITP 2019

24:10

A Verified LL(1) Parser Generator

Theorem parseTableOf_complete :
forall (g : grammar) (tbl : parse_table),
unique_productions g
-> parse_table_correct tbl g
-> exists (tbl’ : parse_table),
ParseTable.Equal tbl tbl’
/\ parseTableOf g = inr tbl’.
In both theorems, the proposition parse_table_correct tbl g says that tbl contains
all and only the lookahead facts about g. It is the mechanized notion of LL(1) parse table
correctness from Section 2.4; the only difference is that in the development, we store an entire
production and its semantic action in each table cell, rather than just the right-hand side.
In the completeness theorem, the unique_productions condition says that the grammar
contains no duplicate productions. Productions are considered duplicates if they are equal up
to their semantic actions—i.e., the unique_productions definition ignores actions. Duplicate
productions always indicate user error; to understand why, consider a grammar with two
productions, X → γ {f } and X → γ {g}. If f and g are the same function, then the
productions are redundant, and one of them can be removed without affecting the grammar’s
semantics. If f and g are different, then the grammar is ambiguous; the parser performs a
single semantic action upon reducing a production, and it is unclear whether that action should
be f or g. Coq functions cannot be compared for equality, so parseTableOf cannot determine
whether duplicate productions are redundant or ambiguous. The unique_productions
property is decidable, however, so the function checks its input grammar for this property
and alerts the user when the check fails. The user can then correct the error in the grammar.
The completeness theorem’s conclusion may seem odd; why don’t we use this version?
Theorem unprovable_parseTableOf_complete :
forall (g : grammar) (tbl : parse_table),
unique_productions g
-> parse_table_correct tbl g
-> parseTableOf g = inr tbl.
In the development, a parse table is simply a finite map in which keys are row/column pairs
and values are cell contents. We use FMaps, a Coq finite map library, to obtain a map
representation and many useful lemmas about map operations. Two maps defined with this
library that contain identical entries are not definitionally equal in Coq because they might
have different internal representations. Thus, if tbl is a correct LL(1) parse table for g,
we cannot prove that parseTableOf returns tbl itself—only that it returns a table tbl’
containing exactly the same entries as tbl, which should be sufficient for any application.
To summarize our progress so far, we have proved that the parse table generator terminates
on all inputs, and that it produces a correct LL(1) parse table for its input grammar whenever
such a table exists.

4

Parser Correctness and Verification

We now turn to the task of defining and verifying the LL(1) parsing algorithm. Our first goal
is to define a function parse that uses an LL(1) parse table tbl and a symbol s to build a
semantic value for a prefix of the token sequence ts:

Definition parse (tbl : parse_table) (s : symbol) (ts : list token) :
sum parse_failure (symbol_semty s * list token).
(The type symbol_semty s is the type of semantic values for symbol s.) We then wish to
verify that as long as the function’s LL(1) parse table argument is correct for some grammar,
its return value is correct with respect to the grammar’s derivation relation. Below are the
three main parser correctness properties that we prove:
1. (Soundness) – If the parser consumes a token sequence, returning a semantic value v for
v
prefix w and an unparsed suffix r, then s −
→ w | r holds.
2. (Error-Free Termination) – The parser never reaches an error state when applied to a
correct LL(1) parse table.
v
3. (Completeness) – If s −
→ w | r holds, then the parser returns v and r when applied to
symbol s and token sequence w ++ r.

4.1

Parser Structure

Because our parser’s correctness specification is the LL(1) derivation relation, it is natural
to structure the parser in a way that mirrors the relation’s structure. An intuitive way of
doing so is to define two mutually recursive functions, parseSymbol and parseGamma, that
respectively consume a symbol and a list of symbols and return a semantic value and a tuple
of semantic values. However, a naïve attempt at defining these two functions leads to a
violation of Coq’s syntactic guardedness condition, which requires all recursive function calls
to have a structurally decreasing argument. The termination checker is not being overly
conservative—a naïvely defined LL(1) parser might actually fail to terminate on certain
inputs! The reason is that our parse tables are simply finite maps, and it is possible to create
a map that would cause the functions to diverge. For example, consider the singleton map
containing the binding (X, a) 7→ X. Applying the parser to this map and a token sequence
beginning with a would cause it to loop infinitely.
The problem with this table is that it includes a left-recursive entry—an entry that leads
the parser from nonterminal X back to X without consuming any input. Our parser detects
left recursion dynamically by maintaining a set of visited nonterminals that is reset to ∅ when
the parser consumes a token. If the parser reaches a nonterminal that is already present in
the visited set, it halts and returns an error value. In our proof of error-free termination, we
show that the parser never actually returns this “left recursion detected” value as long as it
is applied to a correct LL(1) parse table for some grammar, because a grammar that has
such a table contains no left recursion.
Of course, left recursion is not the only failure case—the parser could also determine
that no input prefix is in the language that it recognizes. In this case, it should provide
some information about why it rejected the input. Therefore, our parser returns one of the
following values:
inr (v, r), where v is a semantic value for a prefix of the input tokens and remainder r
is the unparsed suffix, indicating a successful parse.
inl (Reject m r), where m is an error message and remainder r is the suffix that the
parser was unable to consume.
inl (Error m x r), where m is an error message, x is the nonterminal found to be
left-recursive, and r is the unparsed suffix.

After adding left recursion detection, we still have to convince Coq that parseSymbol
and parseGamma terminate, because their termination metric depends on multiple function
parameters. The token sequence decreases structurally in some recursive calls, while in
others, the visited set grows larger (and therefore, its complement relative to the universe of
grammar nonterminals grow smaller). Coq’s Function and Program commands can often
ease the burden of defining functions with subtle termination conditions; both commands
enable the user to write a function and then provide its termination proof after the fact.
Unfortunately, Function and Program do not support mutually recursive functions that are
defined with a well-founded measure. Therefore, we implement well-founded recursion “by
hand,” mimicking the process that these commands perform automatically. The process
involves the following steps:
1. Define a measure meas that maps arguments of parseSymbol and parseGamma to the
following triple of natural numbers:
(First projection) The length of the token sequence.
(Second projection) The cardinality of the visited set’s complement relative to the set
of all grammar nonterminals.
(Third projection) The size of the function’s “symbolic” argument, which is a symbol in
the case of parseSymbol and a list of symbols in the case of parseGamma. We define the
size of a symbol to be 0 and the size of a list of symbols gamma to be 1 + length gamma.
This choice allows parseGamma to call parseSymbol with an unchanged token sequence
and visited set, and it allows parseGamma to call itself under the same conditions as
long as length gamma decreases.
2. Define a lexicographic ordering triple_lt on triples of natural numbers.
3. Add a proof of the measure value’s accessibility in the triple_lt relation (i.e., a proof
that there are no infinite descending chains from the value in triple_lt) as an extra
function argument.
4. Prove lemmas showing that the size of this accessibility proof decreases on recursive calls.
5. Prove that triple_lt is well-founded so that the parser can be called with any initial
set of arguments.
This process yields functions with the following signatures:
Fixpoint parseSymbol (tbl : parse_table) (s : symbol)
(ts : list token) (vis : NtSet.t)
(a : Acc triple_lt (meas tbl ts vis (Sym_arg s)))
: sum parse_failure
(symbol_semty s * {ts’ & length_lt_eq _ ts’ ts}) ...
with parseGamma (tbl : parse_table) (gamma : list symbol)
(ts : list token) (vis : NtSet.t)
(a : Acc triple_lt (meas tbl ts vis (Gamma_arg gamma)))
: sum parse_failure
(rhs_semty gamma * {ts’ & length_lt_eq _ ts’ ts}) ...
In each return type, {ts’ & length_lt_eq _ ts’ ts} is the dependent type of a token
sequence ts’ that is either shorter than the ts argument or definitionally equal to ts. By
including this information in the functions’ dependent return types, we avoid computing the
length of the remaining token sequence at runtime, which would hamper performance.
Finally, we define parse, a top-level interface to the parser that invokes parseSymbol
with an empty visited set and an appropriate accessibility proof term, and that strips out
the return value’s dependent component:

Definition parse (tbl : parse_table) (s : symbol) (ts : list token) :
sum parse_failure (symbol_semty s * list token) :=
match parseSymbol tbl s ts NtSet.empty (triple_lt_wf _) with
| inl failure => inl failure
| inr (v, existT _ ts’ _) => inr (v, ts’)
end.

4.2

Parser Soundness

The first parser correctness property that we prove is soundness with respect to the LL(1)
derivation relation. We show that whenever the parser returns a semantic value for a prefix of
its input, the relation sym_derives_prefix (the mechanized version of the Figure 1 symbol
derivation relation) produces the same value for the same prefix:
Theorem parse_sound :
forall (g : grammar) (tbl
(w r : list token)
parse_table_correct tbl
-> parse tbl s (w ++ r)
-> sym_derives_prefix g

: parse_table) (s : symbol)
(v : symbol_semty s),
g
= inr (v, r)
s w v r.

We prove this theorem via a slightly different statement that implies the previous one:
Lemma parseSymbol_sound :
forall g tbl s ts vis Hacc v r Hle,
parse_table_correct tbl g
-> parseSymbol tbl s ts vis Hacc = inr (v, existT _ r Hle)
-> exists w, w ++ r = ts /\ sym_derives_prefix g s w v r.
The main difference between these two properties is that parse_sound uses the append
function (++) to specify exactly how the function divides its input sequence into a parsed
prefix and an unparsed suffix. It is difficult to reason directly about this statement because
there are multiple ways of dividing the input into a prefix and suffix.
The parseSymbol_sound proof relies on yet another lemma that generalizes over both
parseSymbol and parseGamma. The proof of this latter lemma proceeds by nested induction
on the lexicographic components of the functions’ measure. The proof is straightforward by
design; we were careful to define parseSymbol and parseGamma so that the “success” path
through the functions’ recursive calls mirrors the structure of the derivation relation.

4.3

Parser Error-Free Termination

Our next task is to prove that the parser never returns an error value as long as its table
argument is a correct LL(1) parse table for some grammar:
Theorem parse_terminates_without_error :
forall (g : grammar) (tbl : parse_table)
(s : symbol) (ts ts’ : list token)
(m : string) (x : nonterminal),
parse_table_correct tbl g
-> ~ parse tbl s ts = inl (Error m x ts’).

However, it is certainly possible for parseSymbol and parseGamma to return an error value!
For example, they will produce an error when applied to nonterminal X and a visited set
that already contains X. To prove the top-level function parse safe, we need to specify the
conditions that cause the underlying functions to produce an error, and then prove that
these conditions do not apply to the top-level call.
One error condition is when the parser is applied to symbol s and its visited set already
contains a nonterminal that is reachable from s without any input being consumed. We
formalize this notion of “null-reachability” in the inductive predicate nullable_path:
Inductive nullable_path (g : grammar) (la : lookahead) :
symbol -> symbol -> Prop :=
| DirectPath : forall x z gamma f pre suf,
In (existT _ (x, gamma) f) g.(prods)
-> gamma = pre ++ NT z :: suf
-> nullable_gamma g pre
-> lookahead_for la x gamma g
-> nullable_path g la (NT x) (NT z)
| IndirectPath : forall x y z gamma f pre suf,
In (existT _ (x, gamma) f) g.(prods)
-> gamma = pre ++ NT y :: suf
-> nullable_gamma g pre
-> lookahead_for la x gamma g
-> nullable_path g la (NT y) (NT z)
-> nullable_path g la (NT x) (NT z).
When this predicate holds of two symbols s and s’, there exists a sequence of steps through
the grammar from s to s’ in which all symbols visited along the way are nullable.
The second error condition is when the grammar contains a left-recursive nonterminal,
which is just a special case of null-reachability:
(* symbol s is left-recursive in grammar g on lookahead token la *)
Definition left_recursive (g : grammar) (s : symbol) (la : lookahead) :=
nullable_path g la s s.
We prove a lemma stating that when parseSymbol or parseGamma returns an error value,
one or both of these error conditions holds. The first condition does not apply to parse
because the top-level function calls parseSymbol with an empty visited set. To prove that
the second condition does not apply, we show that a grammar with a correct LL(1) parse
table contains no left recursion. Although standard references mention this property in
passing, we could not find a rigorous proof in the literature. Our proof involves a fair amount
of machinery; it consists of the following steps:
(1) We define sized versions of the nullable_sym (Figure 2) and first_sym (Figure 3)
relations. These versions include a natural number representing the proof term’s size.
(2) We prove that these sizes are deterministic for an LL(1) grammar—any two proofs of
the same nullable_sym or first_sym fact have the same size.
(3) We show that if grammar g contains a left-recursive nonterminal, then there are two
proofs of the same nullable_sym or first_sym fact about g with different sizes.
These steps enable us to prove the lemma LL1_parse_table_impl_no_left_recursion by
obtaining a contradiction from (2) and (3):

Lemma LL1_parse_table_impl_no_left_recursion :
forall (g : grammar) (tbl : parse_table)
(x : nonterminal) (la : lookahead),
parse_table_correct tbl g
-> ~ left_recursive g (NT x) la.

4.4

Parser Completeness

Finally, we prove that our parser is complete—if a grammar symbol derives a semantic value
for a prefix of a token sequence, then the parser produces the same value for the same prefix:
Theorem parse_complete :
forall (g : grammar) (tbl : parse_table)
(s : symbol) (w r : list token)
(v : symbol_semty s),
parse_table_correct tbl g
-> sym_derives_prefix g s w v r
-> parse tbl s (w ++ r) = inr (v, r).
Our error-free termination result simplifies the task of proving completeness. We begin
by proving a more general lemma stating that when a grammar derivation exists, the parser
either returns an error or produces the semantic value from the derivation:
Theorem parseSymbol_error_or_complete :
forall g tbl s w r v vis a,
parse_table_correct tbl g
-> sym_derives_prefix g s w v r
-> (exists m x ts’,
parseSymbol tbl s (w ++ r) vis a = inl (Error m x ts’))
\/ (exists Hle,
parseSymbol tbl s (w ++ r) vis a = inr (v, existT _ r Hle)).
We prove this lemma by induction on the derivation relation, use the error-free termination
theorem to rule out the left disjunct, and use the right disjunct to prove the completeness
theorem itself.

5

Evaluation

To evaluate the efficiency of our generated parsers, we extracted Vermillion to OCaml source
code and generated an LL(1) parser for the JSON data format. We also used Menhir,
a popular OCaml LR(1) parser generator, to produce an unverified parser for the same
grammar and compared the two parsers’ performance on a JSON data set.
We based our Menhir lexer1 and grammar on the ones described in the Real World OCaml
textbook’s tutorial on JSON parsing [12]. We then replicated the grammar in Vermillion’s
input format. Because our tool consumes a list of tokens, we used Menhir to generate a
second parser that acts as a preprocessor for Vermillion—it simply tokenizes an entire JSON
string. In our evaluation, we count this tokenizer’s execution time as part of the LL(1)
parser’s total execution time.
1

The lexer does not support Unicode escape sequences, but nothing prevents Vermillion or Menhir from
handling Unicode tokens in principle.

Figure 8 Average execution times of Menhir and Vermillion JSON parsers.

We ran both JSON parsers on a small data set, averaging the execution times of ten trials
for each data point. The results appear in Figure 8. The Vermillion parser is between two and
four times slower than the unverified Menhir parser on each data point. This comparison is not
entirely scientific, because Menhir and Vermillion use two different parsing algorithms—LR(1)
and LL(1), respectively. Nevertheless, it suggests that Vermillion’s performance is reasonable,
given that it was designed with ease of verification (rather than optimal performance) in
mind. Other certified parsers obtain similar performance results; a validated LR(1) parser [8]
runs about five times slower than its unvalidated counterpart, and a verified PEG interpreter
[9] is two to three times slower than an unverified version.
As an interesting side note, when we first extracted Vermillion to OCaml, we discovered
that its performance was superlinear! This earlier version of the parser periodically computed
the length of the remaining input to determine whether a previous recursive call had consumed
any tokens, and thus whether it was safe to empty the set of visited nonterminals. With some
refactoring, we were able to lift this reasoning about input length into the proof component
of the parser’s dependent return type, ensuring that it is erased at extraction time.

6

Related Work

Barthwal and Norrish [2] use the HOL4 proof assistant to prove the soundness and completeness of generated SLR parsers. Like us, they structure their tool as a generator and a parse
function parameterized by the generator’s output. The parsers are not proved to terminate
on invalid inputs. The work does not include performance results, but the parsers are not
designed to be performant; they compute DFA states during execution rather than statically.
Jourdan et al. [8] present a validator that determines whether a generated LR(1) parser
is sound and complete. A posteriori validation is a flexible and lightweight alternative to full
verification; the validator is compatible with untrusted generators, and its formalization is
small. The validator does not guarantee that a parser terminates on invalid inputs. While
LR(1) parsers are compatible with a larger class of grammars than LL(1) parsers, they often
produce less intuitive error messages.
Parsing Expression Grammars (PEGs) [4] are a language representation that is sometimes
used in place of context-free grammars to specify parsers. Koprowski and Binsztok [9] verify
the soundness and completeness of a PEG parser interpreter. They also ensure that the

interpreter terminates on both valid and invalid inputs by rejecting grammars that fail a
syntactic check for left recursion. Wisnesky et al. [20] verify an optimized PEG parser using
the Ynot framework. Ynot is a library for proving the partial correctness of imperative
programs, so the parser is not guaranteed to terminate. One drawback of using PEG parsers
is that they make greedy choices at decision points—e.g., the rule S → a | ab applied to string
ab parses a instead of ab—which can produce difficult-to-debug behavior.

7

Conclusions

We have verified that our parser generator produces a sound and complete LL(1) parser for
its input grammar whenever such a parser exists, and that the generated parsers terminate
on valid and invalid inputs without using fuel. Below, we discuss two possible extensions of
this work: ruling out parser errors a priori and generating parser source code.
Our parser includes branches that represent error states. These branches survive the
extraction process and slow down the resulting code, even though we prove that the algorithm
never reaches them when applied to a correct LL(1) parse table. An anonymous reviewer
made a useful analogy between the parser and an interpreter that checks for type errors
dynamically, even when a static type system ensures that a valid input program never
triggers these errors—i.e., that “well-typed programs cannot ‘go wrong’ ” [11]. The reviewer
also noted that it might be possible to remove these branches from the parser by making
it a function over correct LL(1) parse tables instead of simply-typed tables, just as one
can remove dynamic type-checking from an interpreter by parameterizing it with typing
derivations instead of raw terms. We chose to rule out errors a posteriori because it is often
simpler to separate the concerns of programming and proving, but the a priori approach
would be more elegant to some observers and certainly more efficient. We hope to explore
the idea in future extensions to this work.
Our parsers represent tables as finite maps and perform map lookups at decision points,
which is a likely source of inefficiency. Many production-grade parser generators produce
source code that is specialized to their input grammar. These parsers represent table lookups
with source-level constructs (e.g., match expressions) instead of data structure operations.
Generated parser code is likely to be more efficient than a table-based interpreter; for
example, Menhir enables the user to choose between these two representations, and an
informal benchmark finds that code generation produces parsers that are two to five times
faster than their table-based counterparts [17]. We could develop a version of our tool that
generates abstract syntax for a language with mechanized semantics, such as Clight [3], and
verify that the abstract syntax representation of a parser is extensionally equivalent to a
table-based parser for the same grammar.

